

NAG Fortran Library Routine Document

G05RBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RBF sets up a reference vector and generates an array of pseudo-random numbers from a Student's t copula with ν degrees of freedom and covariance matrix $\frac{\nu}{\nu-2}C$.

2 Specification

```

SUBROUTINE G05RBF (MODE, DF, M, C, LDC, N, X, LDX, IGEN, ISEED, R, LR,
1 IFAIL)
INTEGER          MODE, DF, M, LDC, N, LDX, IGEN, ISEED(4), LR, IFAIL
double precision C(LDC,M), X(LDX,M), R(LR)

```

3 Description

The Student's t copula, c , is defined by

$$c(u_1, u_2, \dots, u_m; C) = T_{\nu, C}^m(t_{\nu, C_{11}}^{-1}(u_1), t_{\nu, C_{22}}^{-1}(u_2), \dots, t_{\nu, C_{mm}}^{-1}(u_m))$$

where m is the number of dimensions, $T_{\nu, C}^m$ is the multivariate Student's t density function with ν degrees of freedom, mean zero and covariance matrix $\frac{\nu}{\nu-2}C$ and $t_{\nu, C_{ii}}^{-1}$ is the inverse of the univariate Student's t density function with ν degrees of freedom, zero mean and variance $\frac{\nu}{\nu-2}C_{ii}$.

Routine G05LXF is used to generate a vector from a multivariate Student's t distribution and routine G01EBF is used to convert each element of that vector into a uniformly distributed value between zero and one.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05RBF.

4 References

Nelsen R B (1998) *An Introduction to Copulas. Lecture Notes in Statistics 139* Springer

Sklar A (1973) Random Variables: Joint Distribution Functions and Copulas *Kybernetika* **9** 499–460

5 Parameters

1: MODE – INTEGER *Input*

On entry: selects the operation to be performed:

MODE = 0

Initialize and generate random numbers.

MODE = 1

Initialize only (i.e., set up reference vector).

MODE = 2

Generate random numbers using previously set up reference vector.

Constraint: $0 \leq \text{MODE} \leq 2$.

- 2: DF – INTEGER *Input*
On entry: ν , the number of degrees of freedom of the distribution.
Constraint: $DF \geq 3$.
- 3: M – INTEGER *Input*
On entry: m , the number of dimensions of the distribution.
Constraint: $M > 0$.
- 4: C(LDC,M) – **double precision** array *Input*
On entry: matrix which, along with DF defines the covariance of the distribution. Only the upper triangle need be set.
- 5: LDC – INTEGER *Input*
On entry: the first dimension of the array C as declared in the (sub)program from which G05RBF is called.
Constraint: $LDC \geq M$.
- 6: N – INTEGER *Input*
On entry: n , the number of random variates required.
Constraint: $N \geq 1$.
- 7: X(LDX,M) – **double precision** array *Output*
On exit: the array of pseudo-random multivariate Student's t vectors generated by the routine.
- 8: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G05RBF is called.
Constraint: $LDX \geq N$.
- 9: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialization by a prior call to one of the routines G05KBF or G05KCF.
- 10: ISEED(4) – INTEGER array *Input/Output*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 11: R(LR) – **double precision** array *Input/Output*
On entry: if $MODE = 2$, the reference vector as set up by G05RBF in a previous call with $MODE = 0$ or 1 .
On exit: if $MODE = 0$ or 1 , the reference vector that can be used in subsequent calls to G05RBF with $MODE = 2$.
- 12: LR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05RBF is called. If $MODE = 2$, it must be the same as the value of LR specified in the prior call to G05RBF with $MODE = 0$ or 1 .
Constraint: $LR > M(M + 1) + 1$.

13: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1 . Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $\text{MODE} \neq 0, 1 \text{ or } 2$.

IFAIL = 2

On entry, $\text{DF} \leq 2$.

IFAIL = 3

On entry, $\text{M} < 1$.

IFAIL = 4

The covariance matrix C is not positive semi-definite to machine precision.

IFAIL = 5

On entry, $\text{LDC} < \text{M}$.

IFAIL = 6

On entry, $\text{N} < 1$.

IFAIL = 8

On entry, $\text{LDX} < \text{N}$.

IFAIL = 9

On entry, invalid value for IGEN. IGEN must be the same as the value as specified in the prior call to G05RBF with $\text{MODE} = 0$ or 1 .

IFAIL = 11

The reference vector R has been corrupted or M has changed since R was set up in a previous call with $\text{MODE} = 0$ or 1 .

IFAIL = 12

On entry, $\text{LR} < \text{M}(\text{M} + 1)$.

7 Accuracy

See Section 7 of the document for G05LXF for an indication of the accuracy of the underlying multivariate t -distribution.

8 Further Comments

None.

9 Example

The example program prints ten pseudo-random observations from a Student's t copula with ten degrees of freedom and C matrix

$$\begin{bmatrix} 1.69 & 0.39 & -1.86 & 0.07 \\ 0.39 & 98.01 & -7.07 & -0.71 \\ -1.86 & -7.07 & 11.56 & 0.03 \\ 0.07 & -0.71 & 0.03 & 0.01 \end{bmatrix},$$

generated by G05RBF. All ten observations are generated by a single call to G05RBF with $\text{MODE} = 0$. The random number generator is initialized by G05KBF.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05RBF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER      NOUT, LDC, LDX, LR
      PARAMETER    (NOUT=6,LDC=5,LDX=100,LR=LDC*LDC+LDC+2)
*      .. Local Scalars ..
      INTEGER      DF, I, IFAIL, IGEN, J, M, N
*      .. Local Arrays ..
      DOUBLE PRECISION C(LDC,LDC), R(LR), X(LDX,LDC)
      INTEGER      ISEED(4)
*      .. External Subroutines ..
      EXTERNAL     G05KBF, G05RBF
*      .. Executable Statements ..
      CONTINUE

      WRITE (NOUT,*) 'G05RBF Example Program Results'
      WRITE (NOUT,*)

*      Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355

*      Choose the random generator to use
      IGEN = 1

*      Initialise the random generator
      CALL G05KBF(IGEN,ISEED)

*      Set the number of variables and variates
      M = 4
      N = 10

*      Input the upper triangle portion of the covariance matrix
      C(1,1) = 1.69D0
      C(1,2) = 0.39D0
      C(1,3) = -1.86D0
      C(1,4) = 0.07D0
```

```

      C(2,2) = 98.01D0
      C(2,3) = -7.07D0
      C(2,4) = -0.71D0
      C(3,3) = 11.56D0
      C(3,4) = 0.03D0
      C(4,4) = 0.01D0

*      Set the degrees of freedom
      DF = 10

      IFAIL = 0

*      Set up reference vector and generate N numbers
      CALL G05RBF(0,DF,M,C,LDC,N,X,LDX,IGEN,ISEED,R,LR,IFAIL)

*      Display the results
      DO 20 I = 1, N
        WRITE (NOUT,99999) (X(I,J),J=1,M)
20    CONTINUE

      STOP

99999 FORMAT (1X,10F10.4)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05RBF Example Program Results

0.9313	0.2383	0.1422	0.8744
0.2680	0.6898	0.5245	0.1360
0.4794	0.8201	0.8374	0.4155
0.1090	0.4704	0.7063	0.2704
0.4188	0.1213	0.8052	0.7113
0.5303	0.8660	0.4552	0.2842
0.2386	0.3615	0.9266	0.5123
0.7943	0.7209	0.0093	0.3239
0.8986	0.0743	0.5648	0.9152
0.2385	0.0333	0.8660	0.8669
